

## AFFICHEUR LCD – MESURE DE TEMPÉRATURE

### L'afficheur LCD

LCD signifie « *Liquid Crystal Display* », ou en français « *écran à cristaux liquides* ».

On trouve principalement des afficheurs 2x16 ou 4x20, c'est-à-dire à 2 lignes x 16 colonnes, ou 4 lignes x 20 colonnes.


Un tel composant possède 16 broches :

**VSS** (masse, ou *gnd*), **VDD** (+5V), **V0** (réglage du contraste), **RS** (sélection du registre), **RW** (lecture ou écriture), **E** (entrée de validation), **D0 à D7** (bits de données), **A** (anode du rétroéclairage : +5V), **K** (cathode du rétroéclairage : masse).

11 broches seront utilisées :

- VSS et VDD pour l'alimentation ;
- RS et E pour les commandes d'écriture ;
- D4, D5, D6 et D7 pour les bits de données (on parle de communication sur 4 bits) ;
- V0, A et K pour le rétroéclairage (la résistance de 2,2 kΩ peut être remplacée par un potentiomètre pour faire varier la luminosité).

RW sera mis à la masse pour utiliser uniquement l'écriture sur l'afficheur, et les broches D0 à D3 ne seront pas reliées (elles peuvent aussi être mises à la masse).



### Le capteur de température TMP36

Ce composant permet de mesurer la température ambiante. Il est très simple d'utilisation :


L'alimentation utilise les broches extérieures, et une tension de sortie qui varie avec la température est récupérée sur la broche centrale.

L'alimentation peut être comprise entre **2,7 V** et **5,5 V**. Les températures mesurées sont comprises entre **-55°C** et **150°C**.

Le 0°C correspond à une tension de sortie de 500 mV = 0,5 V.

Un changement de 1°C correspond à un changement de 10 mV = 0,01 V sur la tension de sortie.

Ainsi, la formule pour avoir la température est : 
$$\text{Température}(^{\circ}\text{C}) = \frac{\text{Tension\_de\_sortie\_en\_V} \times 1000 - 500}{10}$$




### Montage

Sortez votre matériel et câblez le circuit suivant :

Matériel nécessaire :

- Arduino
- Plaque d'essai
- 1 afficheur LCD
- 1 capteur de température TMP36
- 1 résistance de 100 Ω
- 1 résistance de 2,2 kΩ
- 17 câbles



## Programme

Branchez l'Arduino au câble USB et téléversez le croquis ci-dessous.

### CODE ARDUINO

```
// Croquis P10-LCD-TEMP36.ino
//----- LIBRAIRIE ADDITIONNELLE -----
#include <LiquidCrystal.h> // Librairie pour l'utilisation de l'afficheur LCD
//----- CONSTANTES -----
// création de l'objet lcd (notre afficheur LCD)
LiquidCrystal lcd(8, 9, 10, 11, 12, 13); // RS=8, RW=GND, EN=9, D4=10, D5=11, D6=12, D7=13
#define TMP 5 // Broche de sortie du tmp36 sur la ligne analogique 5 (A5)
//----- VARIABLES -----
int tensionLue; // valeur lue sur la broche de sortie du tmp36
//----- PROCEDURE D'INITIALISATION -----
void setup() {
  lcd.begin(16, 2); // déclaration du LCD à 16 caractères sur 2 lignes
  lcd.clear();      // Effacement du LCD
  // Message de présentation sur 2 lignes durant 3 secondes
  lcd.setCursor(0, 0); // placement : 1ere colonne, 1ere ligne
  lcd.print(" CLG ST GILBERT "); // on écrit...
  lcd.setCursor(0, 1); // placement : 1ere colonne, 2eme ligne
  lcd.print(" LCD - ARDUINO ");
  delay (3000);
  lcd.clear();      // Effacement du LCD
  // Caractères qui ne seront pas effacés
  lcd.setCursor(1, 0);
  lcd.print("Temperature :");
  lcd.setCursor(5, 1);
  lcd.print(" degres C");
}
//----- BOUCLE PRINCIPALE -----
void loop(){
  tensionLue = analogRead(TMP); // Lecture du TMP36, nombre compris entre 0 et 1023
  // conversion en une valeur comprise entre 0.0V et 5.0V (tension réelle)
  float tensionSortie = (tensionLue * 5.0) / 1024.0;
  // Conversion en une température en degrés C
  float temperature = ((tensionSortie * 1000) - 500) / 10;
  // Ecriture de la température pendant 3 secondes
  lcd.setCursor(0, 1); // placement : 1ere colonne, 2eme ligne
  lcd.print(temperature);
  delay(3000);
  lcd.setCursor(0, 1); // placement : 1ere colonne, 2eme ligne
  lcd.print(" "); // Effacement de la température (5 colonnes)
}
```



## Applications

1. Il est possible de définir de nouveaux caractères, par exemple le symbole °. Avant la boucle `setup`, rajouter les lignes ci-contre.

Dans la boucle `setup`, rajouter la ligne :

```
lcd.createChar( 0, degres );
```

et pour afficher le caractère °, il faudra écrire :

```
lcd.write(byte(0));
```

Modifiez le croquis en conséquence.

2. En étudiant bien l'exemple précédent, amusez-vous à créer de nouveaux caractères, des caractères accentués par exemple !

### CODE ARDUINO

```
// Definition du caractere degre
byte degres[8] = {
  B00100,
  B01010,
  B00100,
  B00000,
  B00000,
  B00000,
  B00000,
  B00000,
};
```

3. Vous pouvez tester tous les exemples livrés avec la librairie `LiquidCrystal` en allant dans le logiciel arduino : Fichier → Exemples → `LiquidCrystal`.

**Attention**, il faudra remplacer la ligne de déclaration de l'objet `lcd` par :

```
LiquidCrystal lcd(8, 9, 10, 11, 12, 13);.
```

## Dans le monde réel...

On retrouve les afficheurs LCD dans les calculatrices, les montres, les distributeurs de boissons ou d'autres produits, etc.

